# Maintainer: Baptiste Jonglez <baptiste--aur at jonglez dot org>
pkgname='kea-devel-docs'
pkgver=1.5.0
pkgrel=2
pkgdesc='Developer documentation for Kea, a high-performance, extensible DHCP server engine'
arch=('any')
url='https://kea.isc.org'
license=('MPL2')
depends=()
makedepends=('botan' 'boost' 'log4cplus' # Only here to make ./configure happy, not actually used.
             'libxslt' 'docbook-xsl' 'elinks' 'doxygen' 'graphviz')
source=(https://ftp.isc.org/isc/kea/${pkgver}/kea-${pkgver}.tar.gz{,.asc})
sha512sums=('6d6b7407831311ebe37abce382ce77c664015ddbe3e73ec78153a00b301f98af5be52e26ad4febf5ca1e478d2c1844db4c988b241d2700d758e90b077f176ad8'
            'SKIP')
validpgpkeys=('BE0E9748B718253A28BB89FFF1B11BF05CF02E57') # Internet Systems Consortium, Inc. (Signing key, 2017-2018) <codesign@isc.org>

build() {
  cd "${srcdir}/kea-${pkgver}"
  autoreconf --install
  ./configure \
      --prefix='/usr' \
      --sbindir='/usr/bin' \
      --libexecdir='/usr/bin' \
      --sysconfdir='/etc' \
      --localstatedir='/var'
  make -C doc devel
}

package() {
  cd "${srcdir}/kea-${pkgver}"
  install -d "${pkgdir}/usr/share/doc/kea/"
  cp -r doc/html "${pkgdir}/usr/share/doc/kea/devel"
}
